

**Esercizio 1.** Dato un grafo  $G = \langle N, A \rangle$ , diciamo che  $G$  ha una cricca di dimensione  $k$  se c'è un insieme di  $k$  nodi tale che ogni coppia di nodi al suo interno è connessa da un arco. Scrivere un programma ASP che calcoli tutte le cricche di dimensione data (tramite un fatto "clique(k)" - *Esempio*: "clique(3).") di un dato grafo, rappresentato tramite la relazione binaria arc(Nodo1, Nodo2).

**SOLUZIONE:**

```
arc(1,2).
arc(2,3).
arc(3,4).
arc(1,5).
arc(3,5).
arc(2,5).

arc(Y,X) :- arc(X,Y).
node(X) :- arc(_, X).

clique(3).

inClique(X) | outClique(X) :- node(X).

:- inClique(X), notConnected(X).
notConnected(X) :- node(X), inClique(Y), not arc(X,Y), X != Y.

:- clique(N), not #count {X: inClique(X)} = N.
```

**Esercizio 2.** Dato un grafo  $G = \langle N, A \rangle$ , diciamo che  $G$  ha una cricca di dimensione  $k$  se c'è un insieme di  $k$  nodi tale che ogni coppia di nodi al suo interno è connessa da un arco. Scrivere un programma ASP che calcoli la cricca di dimensione massima di un dato grafo, rappresentato tramite la relazione binaria arc(Nodo1, Nodo2).

**SOLUZIONE:**

```
arc(1,2).
arc(2,3).
arc(3,4).
arc(1,5).
arc(3,5).
arc(2,5).
arc(1,3).
arc(3,6).

arc(Y,X) :- arc(X,Y).
node(X) :- arc(_, X).

inClique(X) | outClique(X) :- node(X).
:- inClique(X), inClique(Y), not arc(X,Y), X!=Y.

:~ outClique(X). [1@1, X]
```

**Esercizio 3.** Si consideri il seguente rompicapo. Obiettivo del gioco è completare una griglia di dimensione 4x4 con numeri che vanno da 1 a 9, rispettando le seguenti semplici regole:

- 1) Ogni riga deve contenere numeri tutti diversi tra loro;
- 2) Ogni colonna deve contenere numeri tutti diversi tra loro;
- 3) La somma dei valori in una riga deve essere uguale al valore assegnato per quella riga;
- 4) La somma dei valori in una colonna deve essere uguale al valore assegnato per quella colonna;

5) La somma dei valori sulla diagonale principale deve essere uguale ad un valore assegnato;

Sono inoltre fornite le seguenti indicazioni:

6) Nella prima riga è presente un 5;

7) La terza riga e la terza colonna contengono valori disposti in ordine crescente;

8) Il numero 3 compare esattamente una volta in tutta la griglia.

Si risolva tale rompicapo tramite un opportuno programma.

*Esempio:* Si consideri la seguente griglia 4x 4, con indicati i valori delle somme degli elementi sulle righe (12,14, 30 e 17), sulle colonne (17,15,23,18) e sulla diagonale principale (il 15 nell'angolo in basso).

				12
				14
				30
				17
17	15	23	18	15

la soluzione del rompicapo è:

4	1	2	5	12
5	2	4	3	14
6	7	8	9	30
2	5	9	1	17
17	15	23	18	15

## **SOLUZIONE:**

```
cifra(1..9).
riga(1..4).
col(1..4).
cella(X,Y) :- riga(X), col(Y).
sommaRiga(1,12).
sommaRiga(2,14).
sommaRiga(3,30).
sommaRiga(4,17).
sommaColonna(1,17).
sommaColonna(2,15).
sommaColonna(3,23).
sommaColonna(4,18).
sommaDiag(15).
```

```
in(X,Y,C) | notIn(X,Y,C) :- cella(X,Y), cifra(C).
```

```
% tutte le celle devono essere riempite
piena(X,Y) :- in(X,Y,_).
:- cella(X,Y), not piena(X,Y).
```

```
% ad ogni cella deve essere assegnato un solo valore
:- in(X,Y,C), in(X,Y,C1), C != C1.
```

```
% in una stessa riga tutti numeri diversi
:- in(X,Y,C), in(X,Y1,C), Y != Y1.
% in una stessa colonna tutti numeri diversi
:- in(X,Y,C), in(X1,Y,C), X != X1.
```

```
% la somma dei valori in una riga deve essere uguale al valore assegnato per quella riga
:- sommaRiga(R,S), not #sum{C: in(R,_,C)} = S.
% la somma dei valori in una colonna deve essere uguale al valore assegnato per quella colonna
:- sommaColonna(Col,S), not #sum{C: in(_,Col,C)} = S.
% la somma dei valori sulla diagonale principale deve essere uguale ad un valore assegnato
```

```

:- sommaDiag(S), not #sum{C: in(R,R,C)} = S.

% nella prima riga e' presente un 5
contiene5(R) :- in(R,C,5).
:- not contiene5(1).

% la terza riga e la terza colonna contengono valori disposti in ordine crescente
crescente(3).
:- crescente(R), in(R,Y,C), in(R,Y1,C1), Y<Y1, C>C1.
:- crescente(Col), in(X,Col,C), in(X1,Col,C1), X<X1, C>C1.

% il numero 3 compare esattamente una volta in tutta la griglia
:- not #count{X,Y: in(X,Y,3)} = 1.
% oppure
:-in(X,Y,3),in(X1,Y1,3),X!=X1.
:-in(X,Y,3),in(X1,Y1,3),Y!=Y1.

```

**Esercizio 4.** Sia  $G=\langle V,E \rangle$  un grafo orientato definito tramite i predicati (fatti) `edge(Vertex1,Vertex2,Color)`, `e vertex(N,Color)`, in cui gli archi e i nodi sono colorati con 3 colori differenti (rosso, verde, blu). Stabilire se esiste un sottoinsieme SDIV di  $V$  avente le seguenti proprietà:

1. SDIV è non vuoto.
2. presi due qualunque nodi  $v_1$  e  $v_2$  in SDIV, e dati i loro colori  $c_1$  e  $c_2$ , nessun arco che li connette può essere di colore  $c_1$  o  $c_2$ .
3. la cardinalità (numero di nodi) di SDIV è massima.

### **SOLUZIONE:**

```

edge(1,2,r).
edge(1,3,b).
edge(1,4,g).
vertex(1,b).
vertex(2,g).
vertex(3,r).
vertex(4,b).

in(X) | out(X) :- vertex(X,_).

% mai nodi con un arco che abbia lo stesso colore di uno dei due
:- in(X), in(Y), edge(X,Y,C), vertex(X,C).
:- in(X), in(Y), edge(X,Y,C), vertex(Y,C).

% massimalita
:~ out(X). [1@1,X]

% insieme non vuoto
almenoUno:- in(X).
:- not almenoUno.

```

**Esercizio 5.** L'anno prossimo si terranno le elezioni per la carica di sindaco nella ridente cittadina di Pasticci Terme, e il primo cittadino in carica ci tiene ad essere rieletto. Pertanto, intende cogliere l'occasione della prossima festa del santo patrono per accattivarsi le simpatie dei concittadini. Ha quindi nominato un comitato con lo scopo di organizzare la più bella festa del circondario, stanziando la maggior quantità

possibile di fondi. Si aiuti il comitato organizzatore a decidere come spendere al meglio questi fondi, tenendo in considerazione tutte le richieste del sindaco e i gusti della cittadinanza.

- La festa si sviluppa in 3 serate. Per ciascuna serata deve essere prevista almeno una attrazione per ciascuna delle seguenti categorie: spettacolo, gioco, religione.
- È chiaro che una stessa attrazione non dovrà essere ripresentata in due serate diverse, pena una brutta figura.
- Ogni serata deve durare almeno 5 e non più di 7 ore.
- Il budget è stabilito dal sindaco e non può essere superato.
- I giovani figli dei gentili elettori preferiscono decisamente le attrattive di carattere giocoso: sarebbe pertanto preferibile massimizzare il loro numero nei 3 giorni. Tuttavia, essendo la festa del santo patrono il parroco insiste perché il numero di attrattive di carattere religioso sia superiore a quello di tutte le altre nei 3 giorni. Il peso politico del parroco è superiore a quello dei ragazzini, pertanto la cosa più importante è accontentare lui.

#### MODELLO DEI DATI IN INPUT:

budget(X)	←	l'ammontare del budget
serata(1..3).	←	le tre serate da organizzare
attrazione(ID,Tipo,Durata,Costo)	←	l'insieme delle attrazioni disponibili